

# Idei...

## TRASAREA CHENARELOR STAS

de Ing. Enikő BITAY  
CLUJ-NAPOCA

**A**i vreo idee? Pune-o pe hîrtie! Încearcă s-o aplici! Și vei reuși! Azi, hai s-o facem împreună...

AutoCAD-ul are o mare flexibilitate și adaptabilitate, admitînd modificări și adăugări de comenzi, după necesitățile utilizatorului. Pentru a ne folosi din plin de facilitățile AutoCAD-ului trebuie să ne creăm un "mediu" de lucru. Ce înțelegem prin aceasta? Înțelegem un nucleu în care ne creăm unelte specifice domeniului nostru de specialitate. Astfel, completăm sau modificăm menu-urile de lucru (vezi "Hello CAD\_FANS" nr. 1, pag. 14 și nr. 7, pag. 28), creăm noi tipuri de linii și fonturi de text, forme și hașuri ("Hello CAD\_FANS" nr. 3, pag. 7 ș.u.), creăm blocuri din entități pe care vrem să le refolosim ("Hello CAD\_FANS" nr. 5, pag. 7), scriem rutine AutoLISP care implementează noi comenzi AutoCAD, etc. Toate aceste facilități simplifică procesul de desenare și măresc spectaculos viteza de lucru în domeniul respectiv.

De exemplu, în domeniul mecanic putem crea o bibliotecă ce poate să cuprindă toate standardele, la care putem apela oricînd pentru a selecta obiectul dorit. Acest lucru se poate realiza prin desenarea parametrizată a entităților cu ajutorul unor programe scrise în AutoLISP. Ulterior, obiectele pot fi selectate și generate dintr-un menu (special scris, sau din cel existent).

Pentru început, vom trata o aplicație simplă, de care avem nevoie în mod inerent pentru a trasa chenarul STAS al desenului. În acest scop, scriem un program în AutoLISP care asigură desenarea oricărui format admis de STAS 1-84, luînd în considerare că fiecare poate fi așezat pe lungime sau lățime.

Luînd în studiu lista de dimensiuni a formatelor standardizate și ținînd cont de specificul programării în AutoLISP, implementăm perechile de dimensiuni sub forma unei liste (identificată prin simbolul LST). Determinarea formatului necesar pentru un desen se face extrem de natural: DUPĂ realizarea desenului la dimensiuni rezultate din întinderea elementelor ce-l compun, indicăm două puncte extreme, PT1 în stînga-jos și PT2 în dreapta-sus (conform figurii), care să delimiteze cît mai spațios și mai echilibrat desenul nostru. Pe baza acestor puncte, programul AutoLISP calculează lungimea și lățimea minim necesare încadrării desenului (a,b). Apoi programul caută în lista LST menționată o pereche de atomi imediat superiori acestor valori, luînd în calcul și distanța dintre conturul pentru decuparea copiei și chenar (GR = 5 mm), precum și fișia de îndosariere (DOS = 20 mm). Căutarea se realizează în două trepte: întîi se verifică lățimea și apoi lungimea. În cazul în care pentru o lățime adoptată nu se găsește nici o pereche de valoare suficient de mare pentru

lungime, lucrurile se inversează: se adoptă lungimea și se caută lățimea. Valorile găsite reprezintă mărimea formatului standardizat (aa,bb). Urmărirea perechilor de atomi se face cu variabila contor N, care reprezintă numărul de ordine în lista LST.

Pentru încadrarea desenului în mijlocul chenarului se definește punctul PM (centrul desenului). Va rezulta o mică excentricitate, datorită fișiei de îndosariere, care se va vedea neimportantă. Cunoscînd deja valorile AA, BB și poziția lui PM se calculează pozițiile punctelor de colț ale conturului de decupare a copiei (P1, P2, P3, P4) și se trasează o polilinie (de grosime 0). La fel se procedează cu chenarul propriu-zis, cunoscînd distanța (dată de GR = 5 mm) față de conturul de decupare. Desenarea acestuia se face prin punctele P11, P22, P33 și P44, tot cu ajutorul comenzii PLINE, dar cu o grosime de linie 0.6.

Mai rămîne trasarea fișiei de îndosariere, ce se aplică tuturor formatelor, cu excepția lui A5 (conform STAS). Pentru asta definim o rutină separată, DOSAR, care conține atît calculul cît și execuția.

În final, cu ajutorul funcției DEFUN din AutoLISP definim funcția FORMAT (programul principal), care poate face parte dintre comenzile AutoCAD:

```
(defun C:FORMAT()
  (chenar)
)
```

Cum se vede, ea apelează rutina CHENAR explicată mai sus. Utilizarea prefixului "C:" este obligatorie, aceasta indicînd definirea unei noi comenzi AutoCAD, cum se știe. În acest fel, comanda FORMAT va putea fi dată pe prompterul "Command:" al AutoCAD-ului,

Vă prezentăm un articol primit de la o cititoare: D-na ing. Eniko BITAY, din Cluj. Deși în numărul trecut am avut un material pe aceeași temă (trasarea chenarelor STAS ale desenelor tehnice), considerăm că modul elegant în care D-na Inginer rezolvă aceeași problemă, ca și interesul pe care îl reprezintă problema în sine, merită să stea în atenția Dvs., fie și numai pentru a vă ilustra încă o aplicație AutoLISP. Singura rezervă pe care o avem este legată de faptul că se folosește o polilinie cu grosime nenulă

după încărcarea programului AutoLISP folosind:

```
(load "FRMT")
unde "FRMT" este numele fișierului (cu extensia .LSP) care conține programul AutoLISP.
```

În urma încărcării primiți informația:

```
C:FORMAT
```

care vă arată că ați implementat o nouă comandă în AutoCAD.


Pentru a nu fi deranjați de defilarea comenzilor în josul ecranului, cît și de apariția cruciulițelor de marcaj (blips), înainte de lansarea rutinei CHENAR se salvează valorile curente ale variabilelor de sistem corespunzătoare (în variabilele BLIP și CMDE), care ulterior vor fi restaurate pentru a nu-i deranja utilizatorului ambianța cu care s-a obișnuit.

Programul FRMT.LSP, implementat de mine pe un calculator compatibil IBM AT - 286, este dat alăturat.

Așa cum este dat aici, programul a fost verificat. Dacă întâmpinați probleme, vă sfătuiesc, ca la orice program AutoLISP, să verificați închiderea corectă a parantezelor, confruntînd-o cu originalul de față. Pentru a vedea în ce loc se "poticnește" încărcarea, dați un:

```
(chenar)
```

și răspundeți la ceea ce vi se cere (indicînd punctele limită), dacă ajungeți pînă la cereri



din prima încercare. Veți vedea care este linia cu probleme. Mai puteți apoi să cereți Dvs. returnarea unor variabile, precedându-le numele cu "!" (ca, de exemplu, !lst), mergînd

```
; Trasare chenar
(defun chenar()
  (setq lst (list '(148 210) '(297 210) '(297 420)
    '(420 297) '(420 594) '(420 891) '(420 1189)
    '(420 1485) '(420 1783) '(420 2080) '(594 420)
    '(594 841) '(594 1261) '(594 1682) '(594 2102)
    '(630 297) '(841 297) '(841 594) '(841 1189)
    '(841 1783) '(841 2378) '(891 420) '(1051 297)
    '(1189 420) '(1189 841) '(1189 1682) '(1189 2523)
    '(1261 297) '(1261 594) '(1471 297) '(1485 420)
    '(1682 297) '(1682 594) '(1682 1189) '(1783 420)
    '(1783 841) '(1892 297) '(2080 420) '(2102 594)
    '(2378 841) '(2523 1189)))
  (setq pt1 (getpoint "nColt stinga jos: "))
  (setq pt2 (getpoint "nColt dreapta sus: "))
  (setq b (abs (- (car pt2) (car pt1))))
  (setq a (abs (- (cadr pt2) (cadr pt1))))
  (setq n 0)
  (setq la (car (nth n lst)))
  (setq gr 5)
  (setq dos 20)
  (while (> (+ a (* 2 gr)) la)
    (setq n (+ n 1))
    (setq la (car (nth n lst)))
  )
  (setq aa la)
  (setq lb (cadr (nth n lst)))
  (while (> (+ b gr dos) lb)
    (setq n (+ n 1))
    (setq lb (cadr (nth n lst)))
    (setq la (car (nth n lst)))
  )
  (setq aa la)
  (setq bb lb)
  (setq pmb (- (+ (car pt1) (/ b 2)) (- dos gr)))
  (setq pma (- (cadr pt2) (/ a 2)))
  (setq pm (list pmb pma))

```

```
(setq p1 (list (- (car pm) (/ bb 2)) (- (cadr pm) (/ aa 2))))
(setq p2 (list (+ (car p1) bb) (cadr p1)))
(setq p3 (list (car p2) (+ (cadr p2) aa)))
(setq p4 (list (- (car p3) bb) (cadr p3)))
(command "pline" p1 "w" "0" "" p2 p3 p4 "c")
(setq p11 (list (+ (car p1) gr) (+ (cadr p1) gr)))
(setq p22 (list (+ (car p11) (- bb (* gr 2))) (cadr p11)))
(setq p33 (list (car p22) (+ (cadr p22) (- aa (* gr 2)))))
(setq p44 (list (car p11) (cadr p33)))
(command "pline" p11 "w" "0.6" "" p22 p33 p44 "c")
(if (/= aa 148) (dosar) (command "redraw"))
)
; Margine Indosariere
(defun dosar()
  (setq pd1 (list (+ (car p1) dos) (cadr p1)))
  (setq pd2 (list (car pd1) (+ (cadr pd1) 297)))
  (setq pd3 (list (- (car pd2) dos) (cadr pd2)))
  (command "pline" pd1 "w" "0" "" pd2 pd3 "")
  (setq pm1 (list (car pd1) (+ (cadr pd1) 148.5)))
  (setq pm2 (list (- (car pm1) dos) (cadr pm1)))
  (command "line" pm1 pm2 "")
  (command "redraw")
)
; Program principal
(defun C:FORMAT()
  (setq blip (getvar "BLIPMODE"))
  (setq cmde (getvar "CMDECHO"))
  (setvar "BLIPMODE" 0)
  (setvar "CMDECHO" 0)
  (chenar)
  (command "zoom" "a")
  (setvar "BLIPMODE" blip)
  (setvar "CMDECHO" cmde)
  (princ)
) ; ----- sfirsit -----
```

în jos pînă ce primiți ca retur "nil". Acea variabilă încă nu a fost inițializată! Procedura e-utilă la orice program AutoLISP, cu scop de depanare. ●●